

UPMC/master/info/APS-4I503

Janvier 2015

Syntaxe

# Analyse des Programmes et Sémantiques

Étude d'un *langage de programmation*.

**Ce ne sera pas** apprendre à programmer avec ce langage.

**Ce sera** savoir parler de *tous les programmes* que l'on peut écrire avec ce langage.

Que peut-on dire de *tous les programmes*?

# Analyse des programmes

## Le programme comme *objet* d'étude

Quantitatif : nombre de lignes de code, nombre de classes définies, performances temps/espace

Qualitatif : *typage*, robustesse, complétude fonctionnelle

## Analyse du *code source*

*Nota : tout n'est pas atteignable de cette manière*

## SYNTAXE

Définir la syntaxe = définir la *grammaire* du langage

Définition inductive

Une (méta)syntaxe pour les grammaires : BNF

Analyse code source = Analyse *dirigée par la syntaxe*

# Notre langage

BOPL : un noyau d'un *langage à objets*

Rèf : Palsberg, J. et Schwartzbach, M. I., Object-Oriented Type Systems. Wiley, 1994.

On y ajoutera des *extensions*

**B**asic **O**bject **P**rogramming **L**anguage

# Un programme BOPL

```
program
```

```
  class Point is
```

```
  vars
```

```
    Int x, y ;
```

```
  methods
```

```
    Point add(Point p)
```

```
    let
```

```
      Point ret ;
```

```
    in
```

```
    begin
```

```
      ret := new Point ;
```

```
      ret.x := (self.x) + (p.x) ;
```

```
      ret.y := (self.y) + (p.y) ;
```

```
      return ret
```

```
    end
```

```
end
```

```
let
```

```
  Point p1, p2, p3 ;
```

```
in
```

```
begin
```

```
  p1 := new Point ;
```

```
  p1.x := 1 ;
```

```
  p1.y := 2 ;
```

```
  p2 := new Point ;
```

```
  p2.x := 5 ;
```

```
  p2.y := 10 ;
```

```
  p3 := p1.add(p2) ;
```

```
  writeln(p3)
```

```
end
```

# Structure d'un programme BOPL

- ▶ En tête – *mot clef* `program` :
  - ▶ Déclaration(s) de classe(s) – mots clef `class` et `is` ;
    - ▶ Variables d'instances – `vars`
    - ▶ Méthode(s) – `methods`
      - Signature
      - Variables locales
      - Corps de la méthode
  - ▶ Déclarations de variables (locales) – mots clef `let` et `in` ;
  - ▶ Corps du programme (*main*) – mots clef `begin` et `end`.

Hiérarchie – *indentation* – *arborescence*

# Grammaire 1

## Les programmes BOPL

PROG ::= program CLASSES LOCALS INSTRLIST  
CLASSES ::=  $\varepsilon$  | CLASS CLASSES  
LOCALS ::=  $\varepsilon$  | let VARDECS in  
INSTRLIST ::= begin INSTRSEQ end

### Typographie et éléments de grammaires

- ▶ *symboles non terminaux* (PETITES CAPITALES)
- ▶ *symboles terminaux* (Machine à écrire ou sans sérif)
- ▶ *élément vide* :  $\varepsilon$
- ▶ *(méta)symbole réservés* ( ::= et | )
- ▶ *séparateurs* (esp. tab. et crlf)

Nota : définition *réursive* de CLASSES



# Grammaire 2

## Définition de classe

`CLASS` ::= `class` id EXTENDS `is`  
VARDECLIST METHODLIST

`EXTENDS` ::=  $\varepsilon$  | `extends` CLASSTYPE

`VARDECLIST` ::=  $\varepsilon$  | `vars` VARDECS

`METHODLIST` ::=  $\varepsilon$  | `methods` METHODS

Avec id : ensemble des *identificateurs* (symboles *terminaux*)

# Grammaire 3

## Types/classes et variables

```
CLASSTYPE ::= Obj | Void | Int | Bool | id
VARDECS   ::= VARDEC | VARDECS VARDEC
VARDEC    ::= CLASSTYPE IDS ;
IDS       ::= IDS , id
```

4 types/classes prédéfinies (mots réservés)

définitions récursives des suites (non vides) VARDECS et  
IDS

symboles séparateurs réservés (; et ,)

# Grammaire 4

## Déclarations de méthodes

```
METHODS      ::= METHOD | METHODS METHOD
METHOD         ::= CLASSTYPE id ( ARGDECLIST )
                LOCALS INSTRLIST
ARGDECLIST    ::= ε | ARGDECS
ARGDECS       ::= ARGDEC | ARGDECS ; ARGDEC
ARGDEC        ::= CLASSTYPE id
INSTRSEQ      ::= INSTR | INSTRSEQ ; INSTR
```

Nota : 3 clauses pour la liste des paramètres (possiblement vide et séparateur infix)

# Grammaire 5

## Instructions

```
INSTR  ::= EXP.id ( ARGLIST )
        | id := EXP
        | EXP.id := EXP
        | return EXP
        | writeln EXP
        | if EXP then INSTRLIST else INSTRLIST
        | while EXP do INSTRLIST
```

# Grammaire 6

## Expressions

EXP ::= EXP + TERM | EXP - TERM  
| EXP or TERM  
| EXP.id | EXP.id ( ARGLIST )  
| super.id ( ARGLIST )  
| EXP instanceof CLASSTYPE

TERM ::= TERM \* FACT | TERM / FACT  
| TERM and FACT | FACT

FACT ::= FACT = BASIC | Fact < BASIC | BASIC

BASIC ::= not EXP | num | id | true | false | nil  
| new CLASSTYPE  
| ( EXP )

ARGLIST ::= ε | ARGS

ARGS ::= EXP | ARGS , EXP

Priorité des opérateurs : TERM, FACT, BASIC